
Descripción
Algunos equipos de Exemys permiten correr en el equipo programas escritos por el usuario, haciéndolos
más flexibles y potentes.

El equipo continuará funcionando normalmente mientras corre el script cargado en su memoria.

Operaciones que se pueden realizar desde un script
• Operaciones Matemáticas
• Operaciones Lógicas
• Operaciones de Temporizado
• Lectura de las entradas propias del equipo y de variables Modbus
• Encendido y Apagado de salidas digitales
• Interpretación de datos del Puerto serie
• Publicación/Subscripción MQTT

(Algunas de estas operaciones estaran o no disponbles según el equipo programado)

Introducción

Este programa permite desarrollar, compilar y transferir scripts a equipos Exemys que lo soporten.

Descripción del Software

A continuación se ve una descripción de las funciones de la pantalla principal.

Conexión con el Equipo

Existen tres formas de conectarse, por USB, por LAN/Ethernet y via MW-XF dependiendo del equipo.

Conexión con el Equipo - USB

Primero debe conectar el equipo a la PC y asegurarse de que este desconectado del “GRDconfig”

Luego desplegar el botón de Conectar y elegir "Open USB Connection"

En algunos equipos con puerto USB, como el RMS1-RM y el wRemote-LoRa, debe
desconectar y volver a conectar el puerto USB del equipo si previamente accedió a la consola
de configuración USB.

 

Conexión con el Equipo - LAN/Ethernet

Primero debe conectar el equipo a la misma red que la PC y asegurarse de que tenga configurada una
dirección IP válida como se indica en su manual del usuario.

Luego desplegar el botón de Conectar y elegir "Open Ethernet Connection"

Al hacerlo deberá ver el/los equipo conectados a su red, o podrá elegir el modo "Remote" para indica
manualmente la dirección IP del equipo si este se encuentra en otra red.

Seleccione el que quiera configurar.

Para conectarse deberá escribir la contraseña del equipo. Es la misma que usa para conectarse el MW.

Conexión con el Equipo via MQTT

Tanto el equipo como el Script Programmer deben estar conectados al mismo broker para enviar y recibir
scripts via broker MQTT de manera remota

Configuración en el equipo

Se debe configurar el tópico de base SCR usado para publicar y para subscribirse. Sobre la base el
equipo agregará un texto como se explica mas adelante.

Tópico de configuración - Identificación de cada equipo

El equipo sumara el ClientID al tópico base de configuración. El tópico de suscripción y publicación en el
ejemplo anterior resultante será el siguiente:

Suscribir -> Exemys/C8-F0-9E-1C-8D-6F/script/1/clientID

Publicar -> Exemys/C8-F0-9E-1C-8D-6F/script/0/clientID

El clientID usado por el equipo será el configurado en la sección MQTT.

Configuración en el ScriptProgrammer

Es necesario cargar en el configurador los parámetros del broker al cual nos queremos conectar y el
tópico base a usar para configurar los equipos remotos.

Para ello hay que entrar a “Software Configuration -> MQTT Connection”

El basic topic debe coincidir con el configurado en los equipo.

Conexión remota

Luego desplegar el botón de Conectar y elegir "Open MQTT Connection"

Allí se debe indicar el ClientID del equipo a configurar.  

Una vez establecida la conexión se podrán usar los botones "Download", "Upload", "Start" y "Stop" como
si estuviera conectado localmente al equipo.

Versiones de Script 1, 2 y 3

En el menú "Project", opción "Properties", pestaña "Script" puede seleccionar entre la versión 1, 2 y
3 del script.

La versión 2 se diferencia de la 1 porque permite el doble de variables ya que pueden ir en minúscula o
mayúscula.

Todos estos equipos funcionan con version con versión 2 o 3. Desde la versión 11.0 en adelante se
pasan automáticamente a versión 3. Los programas escritos en versión 3 son idénticos a los de versión
2. Solo ocupan menos espacio en la memoria del equipo.

La versión 2/3 se diferencia de la 1 porque permite el doble de variables ya que pueden ir en minúscula o
mayúscula.

La versión de script será usada por el Script Programmer en dos situaciones, cuando verifique un script
o cuando trate de enviarlo al equipo.

Si al enviar un script la versión seleccionada no es compatible con el equipo destino verá un mensaje
indicando ese error

Carga y descarga de scripts

Una vez conectados al equipo podremos transferir y descargar los scripts.También de podra detener y
poner en marcha.

Edición de scripts

Para desarrollar un programa solo debemos escribir el código en el panel de edición, el entorno cuenta
con ayuda en la escritura de las funciones y las resalta indicando su correcta escritura.

Una vez que terminamos de escribir el código con el botón “Verify”  compilamos el programa y así
podremos ver si este tiene algún error de sintaxis.

Al momento de compilar en la parte inferior nos indicará si el mismo tiene o no errores, si hay un error se
marcará en rojo la línea en donde se encuentra y en la casilla “Error in Code:” nos dirá la línea, si no
hay aparecerá un cartel indicándolo y en “Error in Code: NONE” .         

En la siguiente imagen vemos un programa con un error, en este caso falta un ";".

El punto y coma falta en la llave anterior a la que esta marcada en rojo, esto se debe a que el compilador
nos indica que encontro un caracter que no es el esperado.

En la siguiente imagen vemos un programa sin errores.

Si no tiene errores podemos, transferir el programa al equipo haciendo clic en “Download to device”.
Aparecerá una ventana que nos indica el estado la descarga y luego un cartel que dice si la descarga
fue exitosa o no.

Depuración de scripts

El Script Programmer dispone de dos herramientas para la depuración de los scripts escritos.

Monitoreo de variables

Con esta herramienta puede ver el valor de las variables númericas o de tipo texto mientras el programa
está corriendo. Tambíen puede modificar el valor de las variables para simular condiciones de trabajo del
script.

Una vez que este conectado al equipo vaya al menú "Project", opción "Watch variables" y luego
"Numeric" o "String" según el tipo de variable a monitorea

Envío y recepción de trazas

Con esta herramienta puede enviar textos desde el script al Script Programmer para seguir el
funcionamiento del script. Tambíen pueden enviar textos simular condiciones de trabajo del script.

Una vez que este conectado al equipo vaya al menú "Project", opción "Trace Window"

Compresión de scripts

El equipo tiene un espacio limitado para guardar el script. En casi todos los equipos máximo son 15.000
caracteres. Existen algunos casos en donde esta limitado a 5.000

Si este espacio no es suficiente para su aplicación puede usar la opción de compresión de scripts. Al
hacerlo el Script Programmer quitará todos los comentarios y tabulaciones antes de enviar el programa.

Si quiere conservar los comentarios de su programa guarde una copia en su computadora.

Para habilitar la compresión vaya al menú "Software Configuration ", opción "Compressed
Transfer"

Introducción
El lenguaje de programación Exemys Script es de tipo bucle, esto quiere decir que se ejecuta todo el
código hasta la última línea y luego vuelve a comenzar.

No hay sentencias para crear bucles dentro del programa, por lo que no se puede detener el flujo del
programa en una sección de código. Se asemeja en este sentido a la programación de PLCs, aunque su
sintaxis es mas cercana al C o a Pascal.

Versiones 1, 2 y 3 del script
Existe 3 versiones de script. La versión 2 se diferencia de la 1 porque permite el doble de variables ya
que pueden ir en minúscula o mayúscula. En la versión 1 las variables solo se escriben en minúscula. La
versión 3 ahorra entre un 20 y un 30% es espacio ocupado por el programa en la memoria del equipo,
permitiendo así cargar programas mas largos. Puede notar la diferencia al presionar el botón de verificar.

Variables
Se dispone de 2 tipos de variables, numéricas del tipo “long”  y de texto del tipo “string”.

No es necesario definir las variables ya que se tiene una cantidad fija.

En la versión 2 de script las variable numéricas son 42, de la “a”  a la “u”  y de la “A”  a la “U”  Las de
texto son 10, de la “v”  a la “z”  y de la “V”  a la “Z”  y con una longitud máxima de 100 caracteres cada
una.

Como las variables numéricas son del tipo “long” cualquier operación que arroje un resultado con
decimales se vera truncado.

El valor inicial de las variables numéricas es 0, en las de texto es un string vacío.

Asignar valor a una variable:

Variables numéricas:

                        a = 652;

Variables de texto:

                        v = 'Hola mundo'

Puede observarse que para cargar texto es necesario colocarlo entre comillas simples.

Concatenación de strings:
 

Para concatenar variables solamente debemos colocar una a continuación de la otra, separándolas
por comas.
 
Por ejemplo:
 
            a = 20;
 
     u = 'La temperatura es de ';
 
     v = ' °C';
 

Si queremos formar la frase 'La temperatura es de 20 °C ' y guardarla en otra variable hacemos
lo siguiente:

     w = u, a, v;
 
Otra forma de hacerlo sería:
 
            w = 'La temperatura es de ', a,' °C';
 
La concatenación solo puede hacerse en asignación de textos a variables string y en la función write_str
 
Inserción de valores ASCII en strings:
 
Si se quiere insertar un valor ASCII en string se puede usar el operador $. Después del operador se
debe indicar al código ASCII en decimal. No se permite el ASCII 0.
 
Por ejemplo:
 
            z = 'Hola mundo',$13,$10;
 
La inserción de valores ASCII solo puede hacerse en asignación de textos a variables string y en la
función write_str

Alias de Variables
Desde la versión 6.1 del Script Programmer se pueden crear "alias" de variables para hacer mas facil la
lectura del código.

Este código se puede transferir a TODOS los equipos de Exemys que soporten Script Programmer, ya
que antes de envíar el código al equipo, se reemplazará el alias por la variable correspondiente.

Como los alias se envian como comentarios dentro del código, al leer el script cargado en el equipo, se
volverán a reemplazar las variables por los alias, a no ser que se haya usado la opción de compresión,
en donde se borran los comentarios.

Se recomiendo SIEMPRE mantener una copia de los scripts cargados en los equipos.

Ejemplo:

Antes Ahora

read_io 2,a,1; 
if a>b {
write_io 1,8,1;
}; 
end;

#[a=SP1]; 
#[b=presion]; 
read_io 2,presion,1;
if presion>SP1 {
write_io 1,8,1;
}; 
end;

Si tiene un script ya escrito, puede sumar los comentarios indicando los alias, enviarlo al equipo (sin
comprimir) y luego leerlo. Al hacerlo, todas las variables que tengan alias serán reemplazadas.

Operadores aritméticos
A continuación veremos los operadores que se pueden utilizar en el script, debemos tener en cuenta que
solo toma valores del tipo “long” por lo que cualquier operación que arroje resultados con decimales se
verá truncada, o sea que solamente obtendremos como resultado la parte entera.
 

Operador Significado
= Igual
^ Exponencial

| Or
& And
+ Suma
- Resta
* Multiplicación
/ División
% Resto

 
Por ejemplo:
 
a = 130;
b = a+5;
 
Resultado: La variable b vale 135.
 

Estructura de un programa
Todas las instrucciones finalizan con punto y coma “;” .
 
Es muy importante tener en cuenta que el programa se ejecuta de manera que cuando se alcanza la
ultima sentencia vuelve a comenzar, debido a esta particularidad disponemos de una sentencia llamada
“start” que se ejecuta solo una vez cuando el programa se inicia. Dentro de este bloque podremos
cargar valores iniciales de variables o todo aquello que consideremos adecuado.
 
Todo programa debe terminar con la instrucción “end;” .
 
A continuación veremos la estructura típica de un programa:
 
            start
     {
         a = 10;
     };
     a = a + 1;
     end;
 
Este simple ejemplo muestra como se utiliza la función “start”  para iniciar por única vez a la variable “a”
con el valor 10 y luego incrementarse en 1 constantemente.
Como podemos ver todas las instrucciones que se encuentran después del bloque “start”  y antes del
“end;”  se ejecutaran de manera cíclica.
 
Comentarios:
 
En caso de querer agregar un comentario al programa se de anteponer al texto el carácter “#”  y
finalizarlo con “;” .
 
Por ejemplo:
 
     start
     {
         a = 10;  #Inicia variable a = 10;
     };
     a = a + 1;    #Incrementa variable a;
     end;
 

Funciones de Control de flujo

El control de flujo especifica el orden en que se van a llevar a cabo ciertas líneas del código.
En nuestro caso tenemos 3 funciones para realizar dicho control, el “start”, el “if-else” y el “end” .  
Función “start”
 
Se utiliza para ejecutar líneas de código por única vez en el momento que el programa se inicia.
Las instrucciones a ejecutar se colocan entre llaves y luego de la de cierre se debe colocar “;” .
 
          Sintaxis:

start
{
...;
...;
};

 
Función “if-else”:

 
Esta función se utiliza para la toma de decisiones, si la condición colocada luego del “if”  es valida se
ejecutaran las instrucciones que estén entre llaves debajo de este en caso contrario se ejecutaran las
que estén dentro del “else” .
 
La condición de ejecución puede utilizar los siguientes operadores:

Operador Significado
= Igual
! Distinto
> Mayor
< Menor

 
La condición se escribe dejando un espacio después del “if”

 
Sintaxis:

           

Solo “if” :

     if condición 
     {
        ...;
        ...;
     };  
            
Luego de la llave de cierre del “if”  debe colocarse “;” .

"if-else” :

     if condición 
     {
        ...;
        ...;
     }
     else
     {
        ...;
        ...;
     };  

En este caso el “;”  se coloca después de la llave de cierre del “else” y no después del “if” .

Función “end”

Solamente se utiliza para indicar el fin del programa luego de ejecutarla se vuelve a la primera línea del
código.
 
Sintaxis:
 
            end;
 

Funciones de interfaz (fuentes/destinos)
Estas funciones permiten tomar datos de distintas fuentes como también enviar datos a distintos
destinos.

Función “read_io”

Permite leer parámetros indexados de distintas fuentes, como los valores de canales de entradas
digitales, salidas digitales, entradas analógicas, etc.
 
Se debe indicar la “fuente”  del dato con un número. En caso de que corresponda, con el parámetro
“índice” , le decimos la posición dentro de esa fuente.
El resultado de la lectura se guarda en una variable numérica.
 
Sintaxis:  

read_io fuente,variable_numerica,indice;              

Las fuentes disponibles dependen del equipo donde se corre el script y la versión de script. En un futuro
pueden agregarse nuevas fuentes.
Refiérase a la seeción fuentes/destinos para mas detalles.

Función “write_io”

Permite escribir parámetros indexados en distintos destinos, como los valores de canales de salidas
digitales, de pulsos, etc.
 
S e debe indicar el “destino”  donde escribir con un número. En caso de que corresponda, con el
parámetro “índice” , le decimos la posición dentro de ese destino.
 
El “valor”  a escribir puede ser un número o una variable numérica.
 
Sintaxis:  

write_io destino,indice,valor;                        

Los destinos disponibles dependen del equipo donde se corre el script y la versión de script. En un
futuro pueden agregarse nuevos destinos.
Refiérase a la seeción fuentes/destinos para mas detalles.  

Función “read_str”:

Permite leer strings originados en distintas fuentes, como lo son los datos recibidos por un puerto serie o
por SMS. Para realizar esto se debe indicar la “fuente” con un número.

El resultado de la lectura se guarda en dos variables, una del tipo string y una del tipo numérica, donde
se indica la longitud del string. Si no hay datos este parámetro dará 0.

Sintaxis:

read_str fuente,variable_numerica,variable_string;
Las fuentes disponibles dependen del equipo donde se corre el script y la versión de script. En un futuro
pueden agregarse nuevas fuentes.
Refiérase a la seeción fuentes/destinos para mas detalles.

Función “write_str” :

Permite escribir strings en distintos “destino”, como el puerto serie o enviar SMS. Para realizar esto se
debe indicar el “destino” con un número y el string a enviar.

Sintaxis:

write_str destino,string;.

Los destinos disponibles dependen del equipo donde se corre el script y la versión de script. En un
futuro pueden agregarse nuevos destinos.
Refiérase a la seeción fuentes/destinos para mas detalles

El string puede ser una variable o un string escrito en la función. Esta función permite concatenación
e inclusión de valores ASCII.

Funciones de string
Estas funciones permiten realizar distintas operaciones con las variables del programa del tipo string.

Función “is_equal”

Compara una variable string con un string (texto fijo o variable string). Devuelve 0 si son distintos y 1 si
son iguales.

Sintaxis:

is_equal variable_numerica,variable_string,string;
Ejemplo:

v='APAGAR BOMBA';
is_equal c,v,'APAGAR BOMBA';
if c=1 {
         #Los textos son iguales;
};

Función “finish_with”

Determina si un variable string termina con un string en particular (texto fijo o variable string). Devuelve 0
si es falso o 1 si verdadero

Sintaxis:

finish_with variable_numerica,variable_string,string;
Ejemplo:

v='APAGAR BOMBA';

finish_with c,v,'BOMBA';
if c=1 {
         #El string guardado en v termina en 'BOMBA';
};

Función “begin_with”

Determina si un variable string comienza con un string en particular (texto fijo o variable string). Devuelve
0 si es falso o 1 si verdadero

Sintaxis:

begin_with variable_numerica,variable_string,string;
Ejemplo:

v='APAGAR BOMBA';
begin_with c,v,'APAGAR';
if c=1 {
         #El string guardado en v empieza con 'APAGAR';
};

Función “contains”:

Determina si una variable string contiene un string en particular (texto fijo o variable string). Devuelve 0 si
no lo encuentra o la posición donde lo encontró si lo encuentra.

Sintaxis:

contains variable_numerica,variable_string,string;

Ejemplo:

v='APAGAR BOMBA';
contains c,v,'GAR';
if c>0 {
         #El string guardado en v contiene el texto 'GAR';
};

Función “upper”

Convierte todos los caracteres de una variable string  a mayúsculas, guardando el resultado en la misma
variable.

Sintaxis:

upper variable_string;
Ejemplo:

v='Apagar';
upper v;
#Ahora v es igual a 'APAGAR';

Función “lower”

Convierte todos los caracteres de una variable string  a minúscula, guardando el resultado en la misma
variable.

Sintaxis:

lower variable_string;
Ejemplo:

v='Apagar';
lower v;
#Ahora v es igual a 'apagar';

Función “strlen”

Devuelve en una variable numérica, la longitud de una variable string.

Sintaxis:

strlen variable_numerica,variable_string;
Ejemplo:

v='Apagar';
strlen c,v;
#Ahora c vale 6;

Función “substr”

Devuelve una parte de una variable string dentro de la misma variable

Sintaxis:

substr inicio,fin,variable_string;
Ejemplo:

v='APAGAR BOMBA';
substr 2,3,v;
#Ahora v vale 'PA';

Funciones de conversión
Estas funciones convierten variables de un tipo a otro.

Función “point”

Coloca el punto decimal a una variable numérica y la convierte a string. 
Como parámetros se le pasa, la variable numérica, la variable string y la cantidad de decimales que
queremos colocarle.

Sintaxis:

point variable_string,variable_numerica,decimales;
Ejemplo:

c=123;
point v,c,1;
#Ahora v vale '12.3';          

Función “aton”

Convierte un número dentro de string a una variable numérica. Empieza en el principio del string y

termina en donde encuentra un valor no numérico o el fin del string.

Sintaxis:

aton variable_numerica,variable_string;
Ejemplo:

v='123 RPM';
aton c,v;
#Ahora c vale 123;           

Funciones “day”,”month”,”year”,”hs”,”min”,”sec” y “nday”

Estas funciones convierten un time_stamp a día, mes, año, hora, minuto, segundo y número de día de
la semana, respectivamente.
Como vimos anteriormente, el fecha/hora actual se lee con la función read_io tipo 7 y se guarda en una
variable numérica que representa la cantidad de segundos desde el 01/01/2000.

Sintaxis:

day dia,timestamp;
month mes,timestamp;
year año,timestamp;
hs hora,timestamp;
min minutos,timestamp;
sec segundos,timestamp;
nday dia_semana,timestamp;

La función “nday”  devuelve el número de día de la semana comenzando por el domingo = 0.

Ejemplo:

read_io 7,e,0; #Lee la hora actual en e;
day f,e;
month g,e;
year h,e;
hs i,e;
min j,e;
sec k,e;
#La fecha y hora actual es f/g/h i:j:k;

Funciones matemáticas y lógicas
Estas funciones realizan ciertas operaciones matemáticas especiales.

Funcion “neg”

Se utiliza para negar una variable numérica. La negación se produce a nivel de bit.

Sintaxis:

neg resultado,inicio;
Primero se coloca la variable en la cual se quiere obtener el resultado y luego la variable a negar.

Ejemplo:

a=32323; #7E43h 
neg b,a;

# b vale 4294934972 (FFFF81BC);

Función “sqtr”

Realiza la raíz cuadrada. Como el Exemys Script solo maneja variables enteras de tipo “long”, la parte
decimal del resultado se verá truncada. Para evitar esto se recomienda multiplicar antes de realizar la
operación.

Sintaxis:

sqrt resultado,inicio;
Ejemplo:

a=225;
sqrt b,a;
#Ahora b vale 15;

Función “scale”

Permite escalar una variable utilizando la ecuación de la recta que pasa por 2 puntos.

Sintaxis:

scale resultado,inicio,x0,x1,y0,y1;
Ejemplo: Escalar la señal 4-20mA de la entrada AN1 en un valor de 0 a 500

read_io 2,a,1; #a = AN1;
scale c,a,400,2000,0,500;
#Ahora c tiene el valor escalado;

Funciones de temporizado
Estas funciones permiten controlar el flujo del programa en función de tiempos.

Funciones “timer” y “check_timer”

Con estas funciones podemos definir un periodo de tiempo y ejecutar instrucciones cuando este se
cumpla.
La forma de uso es la siguiente. Primero ejecutamos la función “timer”  a la cual le pasamos como
parámetros una variable numérica y un tiempo en milisegundos. Luego se consulta esta variable con la
función “check_timer” .

Sintaxis:

timer variable_numerica,tiempo_en_milisegundos;
...
check_timer variable_numerica
{
    ...
    ...
};

Como podemos ver, con la función “timer”  definimos el tiempo, y con “check_timer”  consultamos la
variable cargada anteriormente. Cuando se cumple el periodo, las instrucciones colocadas entre llaves
se ejecutan.
Debemos tener en cuenta que una vez cumplido el periodo la condición siempre será verdadera, y si
volvemos a realizar un “check_timer”  se ejecutaran nuevamente las instrucciones. Normalmente se
carga de vuelta la variable dentro del bloque “check_timer”

Nota: Las funciones de temporizado no deben usarse en aplicaciones de precisión de tiempo, ya que el
temporizado puede presentar cierta dispersión.

2026-03-19

Introducción
Con las funciones read_io, write_io, read_str y write_str se puede acceder a múltiples funciones
adicionales de RME2-PT y RME2-TC. En esta sección del manual se listan las distíntas
fuentes/destinos y luego de explica su uso por función.

Listado de fuentes/destinos

Fuente /
Destino Indice Valor R/W Descripción Función Firmware

2 1 a 8 º x 10 read_io Valor temperatura x 10

E/S

10.0

306 1 a 100

º x 10 
-1/0/1/2
º x 10 

ohms/uV x 100
0/1 > C / F

read_io

1 a 8 Temperatura
9 a 16 Estado de alarma
17 a 24 Temperatura
25 a 32 Resistencia de sensor
(RME2-PT)
25 a 32 uV x 100 (RME2-TC)
33 Tipo de grados

10.0

1000 0 0 a 10 read_io Mensajes recibidos pendientes

MQTT

10.0

1001 0 5 ->
Conectado read_io Estado de conexión con el

broker 10.0

1002 0 2 -> Enviado
OK read_io Estado de ultima publicación 10.0

1000 - read_str Lee primer mensaje en cola de
suscripción 10.0

1003 - read_str Tópico del mensaje en cola de
suscripción leido con 1000 10.0

1001 - write_str Carga tópico a publicar 10.0

1002 - write_str Carga payload a publicar y
publica 10.0

5 4 1 -> COM A write_io
Habilita COM N en modo Script
(en start, write_io 5,4,1; COM A
script)

Serie 10.0

402
12

0 write_io Buffer serie envío - Selecciona e
inicia

Serie
Envío

10.0

405 0 write_io Buffer serie envío - Envia datos
binarios cargados con 403/404 10.0

405 - write_str
Buffer serie envío - Carga y
enviar texto (antes seleccionar
buffer)

10.0

402
13

0 a 199 write_io Buffer serie recepción -
Selecciona y posiciona

Serie
Recepción

10.0

405 0 o N write_io Buffer serie recepción - Vacia al
buffer (0 = Todo) 10.0

405 - 0 a 200 read_io
Buffer serie recepción - Cantidad
de bytes  (antes seleccionar
buffer)

10.0

405 - read_str Recibir texto (antes seleccionar
buffer) 10.0

402 3 1 a 1000 write_io
Buffer Modbus - Selecciona y
posiciona
(Auto incremento para escritura)

Esclavo
Modbus 10.0

10^N
-2 -> 0.01

403

1 -1 -> 0.1
0 -> 1
1 -> 10
2 -> 100

write_io Buffers - Configura exponente
para flotantes

Esclavo
MB/Serie 10.0

2

0-> MSB / 1 -
>LSB

0-> MSW / 1 -
>LSW

write_io Buffers - Configura orden
bytes/words MB/Serie 10.0

404

1

Numero según
formato,

exponente y
orden

read/write_io Buffers - Entero 8 bits sin signo Serie 10.0
2 read/write_io Buffers - Entero 8 bits con signo Serie 10.0
3 read/write_io Buffers - Entero 16 bits sin signo MB/Serie 10.0

4 read/write_io Buffers - Entero 16 bits con
signo MB/Serie 10.0

6 read/write_io Buffers - Entero 32 bits con
signo MB/Serie 10.0

7 read/write_io Buffers - Flotante 32 bits MB/Serie 10.0
35 - read/write_strTraces del Script Programmer Traces 10.0

21 1 a 20 0 a
2147483647 read/write_io Memoria no volátil para números Mem. No

volatil

10.0

121 a
125 - read/write_strMemoria no volátil para texto 1 a

5 10.0

Lecturas de entradas

Fuente /
Destino Indice Valor R/W Descripción

2 1 a 8 º x 10 read_io Valor temperatura x 10

306 1 a 100

º x 10 
-1/0/1/2
º x 10 

ohms/uV x 100
0/1 > C / F

read_io

1 a 8 Temperatura
9 a 16 Estado de alarma
17 a 24 Temperatura
25 a 32 Resistencia de sensor (RME2-PT)
25 a 32 uV (RME2-TC)
33 Tipo de grados

La fuente 2 devuelven las temperatura de las distintas entradas. El “índice” indica el numero de entrada
a leer.

La fuente 306 permite acceder a información mas detallada y datos de configuración.

Ejemplo: Leer la temperatura 4 y guardarlo en la variable c

read_io 2,c,4;

Ejemplo: Leer el estado de la temperatura 4 y guardarlo en la variable c

read_io 306,c,12;

MQTT
Desde sl script se pueden hacer publicaciones bajo criterios y formatos particulares, diferentes a los que
se pueden enviar por la configuración estándar.

Ademas, se pueden recibir los mensajes de los tópicos a los cuales se está suscripto.

Fuente
/

Destino
Indice Valor R/W Descripción Función

1000 0 0 a 10 read_io Mensajes recibidos pendientes

MQTT

1001 0 5 -> Conectado read_io Estado de conexión con el broker
1002 0 2 -> Enviado OK read_io Estado de ultima publicación
1000 - read_str Lee primer mensaje en cola de suscripción

1003 - read_str Tópico del mensaje en cola de suscripción
leido con 1000

1001 - write_str Carga tópico a publicar
1002 - write_str Carga payload a publicar y publica

Ejemplo de publicación: Publica cada 10 segundo un numero que se incrementa en cada envío

check_timer t{ 
  a=a+1;
  read_io 1001,g,0; #Estado de conexión con el broker;
  if g = 5{ #Conectado?;
    write_str 1001,'Prueba/Topico'; #Escribo Topico;
    W = '{ "VAN1":"',a,'}';
    write_str 1002,W;
    read_io 1002,h,0; #h contiene el estado del envio;
  };
  timer t,10000;
};
end;

Ejemplo de recepción de mensajes: Recibe un mensaje e identifica a que tópico corresponde. Lo
envía por la consola de Traces. Previamente se deben carga los tópicos en la configuración web del
equipo.

read_io 1000,a,0; #MQTT - Mensajes recibidos pendientes;
if a > 0{
  read_str 1000,o,Y; #Lee primer mensaje en cola de suscripción;
  read_str 1003,o,X; #Tópico del mensaje en cola de suscripción recien
leido;
  write_str 35,'Mensajes de suscripcion a topico: ',X,': ',Y,$13,$10;
};
end;

Puerto Serie - Modo texto
Desde el script se pueden enviar y recibir datos en modo texto. En el RME2 configurar el puerto serie
como NO habilitado para poder accederlo desde el script.

Para enviar y recibir binario leer "Puerto Serie - Modo binario"

Fuente /
Destino Indice Valor R/W Descripción Función

5 4 1 -> COM A write_io Habilita COM N en modo Script (en start,
write_io 5,4,1; COM A script) Serie

402 12 0 write_io Buffer serie envío - Selecciona e inicia Serie
Envío405 - write_str Buffer serie envío - Carga y enviar texto

(antes seleccionar buffer)

402
13

0 a 199 write_io Buffer serie recepción - Selecciona y
posiciona

Serie
Recepción

405 0 o N write_io Buffer serie recepción - Vacia al buffer (0 =
Todo)

405 - 0 a 200 read_io Buffer serie recepción - Cantidad de bytes 
(antes seleccionar buffer)

405 - read_str Recibir texto (antes seleccionar buffer)

Ejemplo de envío: Inicialización de puerto y envío de texto 'HOLA'

start {
  write_io 5,4,1; #Configura COM A en modo script;
};

if a=0 {
  a=1;
  write_io 402,12,0; #Selecciona buffer serie ;
  write_str 405,'HOLA'; #escritura de eco;
};
end;

Ejemplo de recepción: Recibe caracteres de texto hasta que encuentra un 'CR' (ASCII 13) y guarda el
texto en la variable de string v

start {
  write_io 5,4,1; #Configura COM A en modo script;
};

write_io 402,13,0; #Selecciona buffer serie RX;
read_io 405,b,0; #Cant Recibidos;

if b!0 {
  read_str 405,c,z; #Lee texto en buffer;
  if c!0{
    y=$13; contains d,z,y; #Busca CR;
    if d!0{
      v=z; #Guarda en v el texto recibido;
      write_io 402,13,0; #Selecciona buffer serie RX;
      write_io 405,13,d; #borrar todo lo leido hasta el CR;
    };
  };
};
end;

2024-04-23

Puerto Serie - Modo binario
Desde el script se pueden enviar y recibir datos en modo binario. En el RME2 configurar el puerto serie
como NO habilitado para poder accederlo desde el script.

Para enviar y recibir texto leer "Puerto Serie - Modo texto"

Las fuentes/destinos 402, 403, 404 y 405 usados para acceder al buffer pueden ser compartidos por otra
funciones.

Fuente /
Destino Indice Valor R/W Descripción Función

5 4 1 -> COM A write_io Habilita COM N en modo Script (en start,
write_io 5,4,1; COM A script) Serie

402
12

0 write_io Buffer serie envio - Selecciona e inicia Serie
Envío405 0 write_io Buffer serie envio - Envia datos binarios

cargados con 403/404 

402
13

0 a 199 write_io Buffer serie recepción - Selecciona y
posiciona

Serie
Recepción405 0 o N write_io Buffer serie recepción - Vacia al buffer (0 =

Todo)

405 - 0 a 200 read_io Buffer serie recepción - Cantidad de bytes 
(antes seleccionar buffer)

403

1

10^N
-2 -> 0.01
-1 -> 0.1
0 -> 1
1 -> 10
2 -> 100

write_io Buffers - Configura exponente para
flotantes

Buffer binario2

0-> MSB / 1 -
>LSB

0-> MSW / 1 -
>LSW

write_io Buffers - Configura orden bytes/words

404

1
Numero según

formato,
exponente y

orden

read/write_io Buffers - Entero 8 bits sin signo
2 read/write_io Buffers - Entero 8 bits con signo
3 read/write_io Buffers - Entero 16 bits sin signo
4 read/write_io Buffers - Entero 16 bits con signo
6 read/write_io Buffers - Entero 32 bits con signo
7 read/write_io Buffers - Flotante 32 bits

Ejemplo de envío: Inicialización de puerto y envío de una trama binaria con números en distintos
formatos.

start {
  write_io 5,4,1; #Configura COM A en modo script;
};

if a=0 {
  a=1;

  write_io 402,12,0; #Selecciona buffer binario;

  write_io 403,2,0;#MSB;
  write_io 404,4,-3000; #SINT16;
  write_io 404,4,1000; #SINT16;
  write_io 403,2,0;#MSW;
  write_io 404,6,-70000; #SINT32;

  write_io 404,6,70000; #SINT32;
  write_io 403,2,1;#LSW;
  write_io 404,6,-70000; #SINT32;
  write_io 404,6,70000; #SINT32;
  write_io 403,2,1;#LSB;
  write_io 404,4,1000; #SINT16;
  write_io 403,2,0;#MSB;
  write_io 404,4,1000; #SINT16;

  write_io 403,2,0;#MSW;
  write_io 403,1,2;#10^2; #/100; 
  write_io 404,7,123456; #FLOAT32;
  write_io 403,1,-2;#10^-2; #*100;
  write_io 404,7,123456; #FLOAT32;
  write_io 403,2,1;#LSW;
  write_io 403,1,2;#10^2; #/100;
  write_io 404,7,123456; #FLOAT32;
  write_io 403,1,-2;#10^-2; #*100;
  write_io 404,7,123456; #FLOAT32;

  write_io 405,12,0; #Envia datos cargados;
};
end;

Por el puerto serie salen estos bytes (se expresan en hexadecimal)

F4 48 03 E8 FF FE EE 90 00 01 11 70 EE 90 FF FE 11 70 00 01 E8 03 03 E8 44
9A 51 EC 4C 3C 61 00 51 EC 44 9A 61 00 4B 3C
Ejemplo de recepción: Inicialización de puerto y recibe tramas de 39 bytes de largo con valores en
distintos formato. Los datos que se envian se expresan a continuación en hexadecimal.

F4 48 03 E8 FF FE EE 90 00 01 11 70 EE 90 FF FE 11 70 00 01 E8 03 03 E8 44
9A 51 EC 4C 3C 61 00 51 EC 44 9A 61 00 4B 3C
12 34 56 78 90

Lo datos recibidos se guarda en distintas variables del script. Y esos son los resultados

B=-3000 
C=1000
D=-70000
E=70000
F=-70000
G=70000
H=1000
I=1000
J=123456
K=123456
L=123456
M=123456
start {
  write_io 5,4,1; #Configura COM A en modo script;
};
write_io 402,13,1; #Selecciona buffer binario;

read_io 405,A,0; #Lee cuantos bytes hay en el buffer;
if A>39{
  write_io 403,2,0;#MSB/MSW;
  write_io 402,13,1; read_io 404,B,4; #SINT16;
  write_io 402,13,3; read_io 404,C,4; #SINT16;
  write_io 402,13,5; read_io 404,D,6; #SINT32;
  write_io 402,13,9; read_io 404,E,6; #SINT32;

  write_io 403,2,1;#LSB/LSW;
  write_io 402,13,13; read_io 404,F,6; #SINT32;
  write_io 402,13,17; read_io 404,G,6; #SINT32;
  write_io 402,13,21; read_io 404,H,4; #SINT16;
  write_io 403,2,0;#MSB/MSW;
  write_io 402,13,23; read_io 404,I,4; #SINT16; 
  write_io 402,13,25; write_io 403,1,2; read_io 404,J,7; #10^2 /100 FLOAT32;
  write_io 402,13,29; write_io 403,1,-2; read_io 404,K,7; #10^-2 *100
FLOAT32; 
  write_io 403,2,1;#LSB/LSW;
  write_io 402,13,33; write_io 403,1,2; read_io 404,L,7; #10^2 /100 FLOAT32;
  write_io 402,13,37; write_io 403,1,-2; read_io 404,M,7; #10^-2 *100
FLOAT32; 
  write_io 405,13,0; #Vacia todo el buffer de recepcion;
};
end;

Esclavo Modbus Interno
El equipo cuenta con una zona de memoria de 1000 palabras que pueden ser leidas o escritas desde el
script.

Esto permite principalmente hacer cálculos con los valores de las entradas y ponerlos a disposición en el
registro Modbus interno.

Los registros van desde el Holding Register 1001 al 2000 (41001 al 42000)

Las fuentes/destinos 403 y 404 usados para acceder al buffer pueden ser compartidos por otra
funciones.

Despues de elegir el buffer y la posición con el destino 402, la dirección se autoincrementa a medida que
se cargan datos.

Fuente /
Destino Indice Valor R/W Descripción Función

402 3 1 a 1000 write_io Buffer Modbus - Selecciona y posiciona
(Auto incremento para escritura)

Esclavo
Modbus

403

1

10^N
-2 -> 0.01
-1 -> 0.1

0 -> 1
1 -> 10
2 -> 100

write_io Buffers - Configura exponente para
flotantes

2

0-> MSB / 1 -
>LSB

0-> MSW / 1 -
>LSW

write_io Buffers - Configura orden bytes/words

404

3 Numero según
formato,
exponente y
orden.

read/write_io Buffers - Entero 16 bits sin signo
4 read/write_io Buffers - Entero 16 bits con signo
6 read/write_io Buffers - Entero 32 bits con signo
7 read/write_io Buffers - Flotante 32 bits

Ejemplo de escritura: Carga los siguientes valores en registros Modbus desde el 41001

Registro Modbus Valor (orden byte/word) Valor hexadecimal
41001 -3000 (MSB) F448
41002 1000 (MSB) 03E8

41003:4 -70000 (MSW) FFFEEE90
41005:6 70000 (MSW) 00011170
41007:8 -70000 (LSB) EE90FFFE
41009:10 70000 (LSB) 11700001

41011 1000 (LSB) E803

write_io 402,3,1; #Elige buffer Modbus y posiciona en direccion 1 (41001);

write_io 403,2,0;#MSB;
write_io 404,4,-3000; #41001;
write_io 404,4,1000; #41002;
write_io 403,2,0;#MSW;
write_io 404,6,-70000; #41003:4;
write_io 404,6,70000; #41005:6;
write_io 403,2,1;#LSW;
write_io 404,6,-70000; #41007:8;
write_io 404,6,70000; #41009:10;

write_io 403,2,1;#LSB;
write_io 404,4,1000; #41011;

end;

Ejemplo de lectura:Lee los valores de los registros Modbus 41001 a 41011 y los carga en variables
con distintos formatos.

Registro Modbus Valor hexadecimal Valor variable
41001 F448 a=-3000
41002 03E8 b=1000

41003:4 FFFEEE90 c=-70000
41005:6 00011170 d=70000
41007:8 EE90FFFE e=-70000
41009:10 11700001 f=70000

41011 E803 g=1000

write_io 403,2,0;#MSB;
write_io 402,3,1; read_io 404,a,4; #41001;
write_io 402,3,2; read_io 404,b,4; #41002;
write_io 403,2,0;#MSW;
write_io 402,3,3; read_io 404,c,6; #41003:4;
write_io 402,3,5; read_io 404,d,6; #41005:6;
write_io 403,2,1;#LSW;
write_io 402,3,7; read_io 404,e,6; #41007:8;
write_io 402,3,9; read_io 404,f,6; #41009:10;
write_io 403,2,1;#LSB;
write_io 402,3,11; read_io 404,g,4; #41011;

end;

2024-04-23

 

RME2-PT/TC - Programacion de Scripts Esclavo Modbus interno

34 / 36



Envio/Recepción de mensajes al Script Programmer ("Traces")

Fuente /Destino R/W Descripción Función

35 read/write_str Traces del Script Programmer Traces

El destino 35 permite enviar texto a la ventana de "Traces" en el Script Programmer. Esto es
particularmente util cuando se está probando un script nuevo.

Hay una consideración con respecto a los caracteres de espacio y guión bajo. El caracter de espacion
será reemplazado por un guión bajo antes de leer con read_str 35 . El guión bajo será reemplazado por
un espacio luego de enviar con write_str 35 .

Si el Script Programmer no se encuenta conectado al escribir en el destino 35 el texto simplemente se
perderá pero no afectará al funcionamiento del programa.

Lectura/Escritura de memoria no volátil

El equipo puede guardar texto o numeros en memoria no volátil para conservar los valores aunque el
equipo se apague.

Fuente /
Destino Indice Valor R/W Descripción Función

21 1 a 20 0 a 2147483647 read/write_io Memoria no volátil para números Mem. No
volatil121 a

125 - read/write_strMemoria no volátil para texto 1 a 5

Para números

La fuente/destino 21 permite leer y escribir hasta valóres numéricos en la memoria no volátil del equipo.

Ejemplo: Leer el valor numérico almacenado en al posición 15 de la memoria no volátil en la variable g

read_io 21,g,15;

Para texto

Las fuentes/destinos 121 a 125 permiten leer y escribir hasta 5 textos en la memoria no volátil del equipo.

Ejemplo: Escribir la palabra 'hola' en la tercera posición de la memoria no volátil para textos.

write_str 123,'hola';

